

[Previous page](#)[Table of contents](#)[Chapter overview](#)[Next page](#)

Appendix A

Instruction Set

Introduction

Appendix contains all instructions presented separately with examples for their use. Syntax, description and its effects on status bits are given for each instruction.

- [A.1 MOVLW](#)
- [A.2 MOVWF](#)
- [A.3 MOVF](#)
- [A.4 CLRW](#)
- [A.5 CLRF](#)
- [A.6 SWAPF](#)
- [A.7 ADDLW](#)
- [A.8 ADDWF](#)
- [A.9 SUBLW](#)
- [A.10 SUBWF](#)
- [A.11 ANDLW](#)
- [A.12 ANDWF](#)
- [A.13 IORLW](#)
- [A.14 IORWF](#)
- [A.15 XORLW](#)
- [A.16 XORWF](#)
- [A.17 INCF](#)
- [A.18 DECF](#)
- [A.19 RLF](#)
- [A.20 RRF](#)
- [A.21 COMF](#)
- [A.22 BCF](#)
- [A.23 BSF](#)
- [A.24 BTFSC](#)
- [A.25 BTFSS](#)
- [A.26 INCFSZ](#)
- [A.27 DECFSZ](#)
- [A.28 GOTO](#)
- [A.29 CALL](#)
- [A.30 RETURN](#)
- [A.31 RETLW](#)
- [A.32 RETFIE](#)
- [A.33 NOP](#)
- [A.34 CLRWDI](#)

- [A.35 SLEEP](#)

A.1 MOVLW Write constant in W register

Syntax: *[label]* MOVLW **k**
Description: 8-bit constant **k** is written in **W** register.
Operation: $k \Rightarrow (W)$
Operand: $0 \leq k \leq 255$
Flag: -
Number of words: 1
Number of cycles: 1

Example 1 MOVLW 0x5A

After instruction: W=0x5A

Example 2 MOVLW REGISTAR

Before instruction: W=0x10 and REGISTAR=0x40

After instruction: W=0x40

A.2 MOVWF Copy W to f

Syntax: *[label]* MOVWF **f**
Description: Contents of **W** register is copied to **f** register.
Operation: $W \Rightarrow (f)$
Operand: $0 \leq f \leq 127$
Flag: -
Number of words: 1
Number of cycles: 1

Example 1 MOVWF OPTION_REG

Before instruction: OPTION_REG=0x20
 W=0x40

After instruction: OPTION_REG=0x40
 W=0x40

Example 2 MOVWF INDF

Before instruction: W=0x17
 FSR=0xC2
 address contents 0xC2=0x00

After instruction: W=0x17
 FSR=0xC2
 address contents 0xC2=0x17

A.3 MOVF Copy f to d

Syntax: `[label] MOVF f, d`
Description: Contents of **f** register is stored in location determined by **d** operand.
 If **d=0**, destination is **W** register.
 If **d=1**, destination is **f** register itself.
 Option **d=1** is used for testing the contents of **f** register because execution of this instruction affects Z flag in STATUS register.

Operation: $f \Rightarrow (d)$
Operand: $0 \leq f \leq 127$
 $d \in [0, 1]$
Flag: Z
Number of words: 1
Number of cycles: 1

Example 1 `MOVF FSR, 0`

Before instruction: `FSR=0xC2`
`W=0x00`
 After instruction: `W=0xC2`
`Z=0`

Example 2 `MOVF INDF, 0`

Before instruction: `W=0x17`
`FSR=0xC2`
`address contents 0xC2=0x00`
 After instruction: `W=0x17`
`FSR=0xC2`
`address contents 0xC2=0x00`
`Z=1`

A.4 CLRW Write 0 in W

Syntax: `[label] CLRW`
Description: Contents of **W** register evens out to zero, and Z flag in STATUS register is set to one.
Operation: $0 \Rightarrow (W)$
Operand: -
Flag: Z
Number of words: 1
Number of cycles: 1

Example `CLRW`

Before instruction: `W=0x55`
 After instruction: `W=0x00`
`Z=1`

A.5 Write 0 in f

Syntax: `[label] CRLF f`
Description: Contents of '**f**' register evens out to zero, and Z flag in status register is set to one.
Operation: $0 \Rightarrow f$
Operand: $0 \leq f \leq 127$
Flag: Z
Number of words: 1
Number of cycles: 1

Example 1 CRLF STATUS

Before instruction: STATUS=0xC2
 After instruction: STATUS=0x00
 Z=1

Example 2 CLRF INDF

Before instruction: FSR=0xC2
 address contents 0xC2=0x33
 After instruction: FSR=0xC2
 address contents 0xC2=0x00
 Z=1

A.6 SWAPF Copy the nibbles from f to d crosswise

Syntax: `[label] SWAPF f, d`
Description: Upper and lower half of **f** register exchange places.
 If **d=0**, result is stored in **W** register.
 If **d=1**, result is stored in **f** register.
Operation: $f<0:3> \Rightarrow d<4:7>, f<4:7> \Rightarrow d<0:3>;$
Operand: $0 \leq f \leq 127$
 $d \in [0,1]$
Flag: -
Number of words: 1
Number of cycles: 1

Example 1 SWAP REG, 0

Before instruction: REG=0xF3
 After instruction: REG=0xF3
 W=0x3F

Example 2 SWAP REG, 1

Before instruction: REG=0xF3
 After instruction: REG=0x3F

A.7 ADDLW Add W to a constant

Syntax: `[label] ADDLW k`
Description: Contents of **W** register is added to 8-bit constant **k** and result is stored in **W** register.
Operation: $(W) + k \Rightarrow W$
Operand: $0 \leq k \leq 255$
Flag: C, DC, Z
Number of words: 1
Number of cycles: 1

Example 1 `ADDLW 0x15`

Before instruction: `W=0x10`

After instruction: `W=0x25`

Example 2 `ADDLW REG`

Before instruction: `W=0x10`

register contents `REG=0x37`

After instruction: `W=0x47`

A.8 ADDWF Add W to f

Syntax: `[label] ADDWF f, d`
Description: Add contents of register **W** to register **f**.
 If **d=0**, result is stored in **W** register.
 If **d=1**, result is stored in **f** register.
Operation: $(W) + (f) \Rightarrow d$
 $d \in [0,1]$
Operand: $0 \leq f \leq 127$
Flag: C, DC, Z
Number of words: 1
Number of cycles: 1

Example 1 `ADDWF FSR, 0`

Before instruction: `W=0x17`

`FSR=0xC2`

After instruction: `W=0xD9`

`FSR=0xC2`

Example 2 `ADDLW INDF, 1`

Before instruction: `W=0x17`

`FSR=0xC0`

address contents `0xC2=0x20`

After instruction: `W=0x17`

`FSR=0xC2`

address contents `0xC2=0x37`

A.9 SUBLW Subtract W from a constant

Syntax: `[label] SUBLW k`
Description: Contents of **W** register is subtracted from **k** constant, and result is stored in **W** register.
Operation: $k - (W) \Rightarrow W$
Operand: $0 \leq k \leq 255$
Flag: C, DC, Z
Number of words: 1
Number of cycles: 1

Example 1 SUBLW 0x03

Before instruction: W=0x01, C=x, Z=x
 After instruction: W=0x02, C=1, Z=0 Result > 0

Before instruction: W=0x03, C=x, Z=x
 After instruction: W=0x00, C=1, Z=1 Result = 0

Before instruction: W=0x04, C=x, Z=x
 After instruction: W=0xFF, C=0, Z=0 Result < 0

Example 2 SUBLW REG

Before instruction: W=0x10
 contents REG=0x37
 After instruction: W=0x27
 C=1 Result > 0

A.10 SUBWF Subtract W from f

Syntax: `[label] SUBWF f, d`
Description: Contents of **W** register is subtracted from the contents of **f** register.
 If **d=0**, result is stored in **W** register.
 If **d=1**, result is stored in **f** register.
Operation: $(f) - (W) \Rightarrow d$
Operand: $0 \leq f \leq 127$
 $d \in [0,1]$
Flag: C, DC, Z
Number of words: 1
Number of cycles: 1

Example 1 SUBWF REG, 1

Before instruction: REG=3, W=2, C=x, Z=x
 After instruction: REG=1, W=2, C=1, Z=0 Result > 0

Before instruction: REG=2, W=2, C=x, Z=x
 After instruction: REG=0, W=2, C=1, Z=1 Result = 0

Before instruction: REG=1, W=2, C=x, Z=x
 After instruction: REG=0xFF, W=2, C=0, Z=0 Result < 0

A.11 ANDLW Logic AND W with constant

Syntax: `[label] ANDLW k`
Description: Performs operation logic AND over the contents of **W** register and constant **k**.
 Result is stored in **W** register.
Operation: $(W) \text{ .AND. } k \Rightarrow W$
Operand: $0 \leq k \leq 255$
Flag: Z
Number of words: 1
Number of cycles: 1

Example 1 ANDLW 0x5F

Before instruction:	W=0xA3	; 0101 1111 (0x5F)
After instruction:	W=0x03	; 1010 0011 (0xA3)

		; 0000 0011 (0x03)

Example 2 ANDLW REG

Before instruction:	W=0xA3	; 1010 0011 (0xA3)
	REG=0x37	; 0011 0111 (0x37)
After instruction:	W=0x23	-----
		; 0010 0011 (0x23)

A.12 ANDWF Logic AND W with f

Syntax: `[label] ANDWF f, d`
Description: Performs operation of logic AND over the contents of **W** and **f** registers.
 If **d=0**, result is stored in **W** register.
 If **d=1**, result is stored in **f** register.
Operation: $(W) \text{ .AND. } f \Rightarrow d$
Operand: $0 \leq f \leq 127$
 $d \in [0,1]$
Flag: Z
Number of words: 1
Number of cycles: 1

Example 1 ANDWF FSR, 1

Before instruction:	W=0x17, FSR=0xC2	; 0001 0111 (0x17)
After instruction:	W=0x17, FSR=02	; 1100 0010 (0xC2)

		; 0000 0010 (0x02)

Example 2 ANDWF FSR, 0

Before instruction:	W=0x17, FSR=0xC2	; 0001 0111 (0x17)
After instruction:	W=0x02, FSR=0xC2	; 1100 0010 (0xC2)

		; 0000 0010 (0x02)

A.13 IORLW Logic OR W with constant

Syntax: `[label] IORLW k`
Description: Operation logic OR is performed over the contents of **W** register and over 8-bit constant **k**, and result is stored in **W** register.
Operation: $(W) .OR. (k) \Rightarrow W$
Operand: $0 \leq k \leq 255$
Flag: Z
Number of words: 1
Number of cycles: 1

Example 1 `IORLW 0x35`

Before instruction: `W=0x9A`
 After instruction: `W=0xBF`
`Z=0`

Example 2 `IORLW REG`

Before instruction: `W=0x9A`
`content REG=0x37`
 After instruction: `W=0x9F`
`Z=0`

A.14 IORWF Logic OR W with f

Syntax: `[label] IORWF f, d`
Description: Operation logic OR is performed over the contents of **W** and **f** registers.
 If **d=0**, result is stored in **W** register.
 If **d=1**, result is stored in **f** register.
Operation: $(W) .OR. (f) \Rightarrow d$
Operand: $0 \leq f \leq 127$
 $d \in [0,1]$
Flag: Z
Number of words: 1
Number of cycles: 1

Example 1 `IORWF REG, 0`

Before instruction: `REG=0x13, W=0x91`
 After instruction: `REG=0x13, W=0x93`
`Z=0`

Example 2 `IORWF REG, 1`

Before instruction: `REG=0x13, W=0x91`
 After instruction: `REG=0x93, W=0x91`
`Z=0`

A.15 XORLW Logic exclusive OR W with constant

Syntax: `[label] XORLW k`
Description: Operation exclusive OR (XOR) is done over the contents of **W** register and constant **k**, and result is stored in **W** register.
Operation: $(W) \text{ .XOR. } k \Rightarrow W$
Operand: $0 \leq k \leq 255$
Flag: Z
Number of words: 1
Number of cycles: 1

Example 1 XORLW 0xAF

Before instruction:	W=0xB5		; 1010 1111 (0xAF)
After instruction:	W=0x1A		; 1011 0101 (0xB5)

			; 0001 1010 (0x1A)

Example 2 XORLW REG

Before instruction:	W=0xAF		; 1010 1111 (0xA3)
	REG=0x37		; 0011 0111 (0x37)
After instruction:	W=0x18		-----
	Z=0		; 0001 1000 (0x18)

A.16 XORWF **Logic exclusive OR W with f**

Syntax: `[label] XORWF f, d`
Description: Operation exclusive OR is performed over the contents of **W** and **f** registers.
 If **d=0**, result is stored in **W** register.
 If **d=1**, result is stored in **f** register.
Operation: $(W) \text{ .XOR. } (f) \Rightarrow d$
Operand: $0 \leq f \leq 127$
Flag: Z
Number of words: 1
Number of cycles: 1

Example 1 XORWF REG, 1

Before instruction:	REG=0xAF, W=0xB5		; 1010 1111 (0xAF)
After instruction:	REG=0x1A, W=0xB5		; 1011 0101 (0xB5)

			; 0001 1010 (0x1A)

Example 2 XORWF REG, 0

Before instruction:	REG=0xAF, W=0xB5		; 1010 1111 (0xAF)
After instruction:	REG=0xAF, W=0x1A		; 1011 0101 (0xB5)

			; 0001 1010 (0x1A)

A.17 INCF **Increment f**

Syntax: $[label] \text{ INCF } f, d$
Description: Increments f register by one.
 If $d=0$, result is stored in **W** register.
 If $d=1$, result is stored in f register.
Operation: $(f) + 1 \Rightarrow d$
Operand: $0 \leq f \leq 127$
 $d \in [0,1]$
Flag: Z
Number of words: 1
Number of cycles: 1

Example 1 INCF REG, 1

Before instruction: REG=0xFF
 Z=0
 After instruction: REG=0x00
 Z=1

Example 2 INCF REG, 0

Before instruction: REG=0x10
 W=x
 Z=0
 After instruction: REG=0x10
 W=0x11
 Z=0

A.18 DECF Decrement f

Syntax: `[label] DECF f, d`
Description: Decrements `f` register by one.
 If `d=0`, result is stored in **W** register.
 If `d=1`, result is stored in `f` register.
Operation: $(f) - 1 \Rightarrow d$
Operand: $0 \leq f \leq 127$
 $d \in [0,1]$
Flag: Z
Number of words: 1
Number of cycles: 1

Example 1 DECF REG, 1

Before instruction: REG=0x01
 Z=0
 After instruction: REG=0x00
 Z=1

Example 2 DECF REG, 0

Before instruction: REG=0x13
 W=x
 Z=0
 After instruction: REG=0x13
 W=0x12
 Z=0

A.19 RLF Rotate f to the left through CARRY

Syntax: `[label] RLF f, d`

Description: Contents of **f** register is rotated by one space to the left through C (Carry) flag.

If **d=0**, result is stored in **W** register.

If **d=1**, result is stored in **f** register.

Operation: $(f\langle n \rangle) \Rightarrow d\langle n+1 \rangle, f\langle 7 \rangle \Rightarrow C, C \Rightarrow d\langle 0 \rangle;$


Operand: $0 \leq f \leq 127$

d $\in [0,1]$

Flag: C

Number of words: 1

Number of cycles: 1



Example 1 `RLF REG, 0`

Before instruction: REG=1110 0110
C=0

After instruction: REG=1110 0110
W=1100 1100
C=1


Example 2 `RLF REG, 1`

Before instruction: REG=1110 0110
C=0

After instruction: REG=1100 1100
C=1

A.20 RRF Rotate f to the right through CARRY

Syntax:	<code>[label] RRF f, d</code>
Description:	Contents of f register is rotated by one space to the right through C (Carry) flag. If d=0 , result is stored in W register. If d=1 , result is stored in f register.
Operation:	$(f\langle n \rangle) \Rightarrow d\langle n-1 \rangle, f\langle 0 \rangle \Rightarrow C, C \Rightarrow d\langle 7 \rangle;$
Operand:	$0 \leq f \leq 127$ $d \in [0,1]$
Flag:	C
Number of words:	1
Number of cycles:	1

**Example 1** RRF REG, 0

Before instruction: REG=1110 0110
W=x
C=0

After instruction: REG=1110 0110
W=0111 0011
C=0

Example 2 RRF REG, 1

Before instruction: REG=1110 0110
C=0

After instruction: REG=0111 0011
C=0

A.21 COMF Complement f

Syntax: `[label] COMF f, d`
Description: Contents of **f** register is complemented.
 If **d=0**, result is stored in **W** register.
 If **d=1**, result is stored in **f** register.
Operation: $\overline{(f)} \Rightarrow d$
Operand: $0 \leq f \leq 127$
 $d \in [0,1]$
Flag: Z
Number of words: 1
Number of cycles: 1

Example 1 `COMF REG, 0`

Before instruction:	REG=0x13	; 0001 0011 (0x13)
After instruction:	REG=0x13	; complement
	W=0xEC	-----
		; 1110 1100 (0xEC)

Example 2 `COMF INDF, 1`

Before instruction:	FSR=0xC2
	address contents (FSR)=0xAA
After instruction:	FSR=0xC2
	address contents (FSR)=0x55

A.22 BCF Reset bit b in f

Syntax: `[label] BCF f, b`
Description: Reset bit **b** in **f** register.
Operation: $(0) \Rightarrow f[b]$
Operand: $0 \leq f \leq 127$
 $0 \leq b \leq 7$
Flag: -
Number of words: 1
Number of cycles: 1

Example 1 `BCF REG, 7`

Before instruction:	REG=0xC7	; 1100 0111 (0xC7)
After instruction:	REG=0x47	; 0100 0111 (0x47)

Example 2 `BCF INDF, 3`

Before instruction:	W=0x17
	FSR=0xC2
	address contents (FSR)=0x2F
After instruction:	W=0x17
	FSR=0xC2
	address contents (FSR)=0x27

A.23 BSF Set bit b in f

Syntax: `[label] BSF f, b`
Description: Set bit **b** in **f** register.
Operation: $1 \Rightarrow f$
Operand: $0 \leq f \leq 127$
 $0 \leq b \leq 7$
Flag: -
Number of words: 1
Number of cycles: 1

Example 1 BSF REG, 7

Before instruction: REG=0x07 ; 0000 0111 (0x07)
 After instruction: REG=0x17 ; 1000 0111 (0x17)

Example 2 BCF INDF, 3

Before instruction: W=0x17
 FSR=0xC2
 address contents (FSR)=0x20
 After instruction: W=0x17
 FSR=0xC2
 address contents (FSR)=0x28

A.24 BTFSC Test bit b in f, skip if it = 0

Syntax: `[label] BTFSC f, b`
Description: If bit **b** in **f** register equals zero, then we skip the next instruction. If bit **b** equals zero, during execution of the current instruction, execution of the next one is disabled, and NOP instruction executes instead thus making the current one a two-cycle instruction.
Operation: Skip next instruction if $(f)=0$
Operand: $0 \leq f \leq 127$
 $0 \leq b \leq 7$
Flag: -
Number of words: 1
Number of cycles: 1 or 2 depending on a **b** bit

Example

```
LAB_01    BTFSC REG,1    ;Test bit no.1 in REG
LAB_02    . . . . .      ;Skip this line if =0
LAB_03    . . . . .      ;Skip here if =1
```

Before instruction, program counter was at address LAB_01.

After instruction, if the first bit in REG register was zero, program counter points to address LAB_03.

If the first bit in REG register was one, program counter points to address LAB_02.

A.25 BTFSS Test bit b in f, skip if = 1

Syntax:	<code>[label] BTFSS f, b</code>
Description:	If bit b in f register equals one, then skip over the next instruction. If bit b equals one, during execution of the current instruction, the next one is disabled, and NOP instruction is executed instead, thus making the current one a two-cycle instruction.
Operation:	Skip next instruction if $(f \langle b \rangle) = 1$
Operand:	$0 \leq f \leq 127$ $0 \leq b \leq 7$
Flag:	-
Number of words:	1
Number of cycles:	1 or 2 depending on a b bit

Example

```
LAB_01      BTFSS  REG,1      ; Test bit no.1 in REG
LAB_02      . . . . .        ; Skip this line if =1
LAB_03      . . . . .        ; Skip here if =0
```

Before instruction, program counter was at address LAB_01

After instruction, if the first bit in REG register was one, program counter points to address LAB_03.

If the first bit in REG register was zero, program counter points to address LAB_02.

A.26 INCFSZ Increment f, skip if=0

Syntax:	<code>[label] INCFSZ f, d</code>
Description:	Contents of f register is incremented by one. If d=0 , result is stored in W register. If d=1 , result is stored in f register. If result =0, the next instruction is executed as NOP making the current one a two-cycle instruction.
Operation:	$(f) + 1 \Rightarrow d$
Operand:	$0 \leq f \leq 127$ $d \in [0,1]$
Flag:	-
Number of words:	1
Number of cycles:	1 or 2 depending on a result

Example

```
LAB_01      INCFSZ REG, 1      ; Increase the contents REG by one.
LAB_02      . . . . .        ; Skip this line if =0
LAB_03      . . . . .        ; Skip here if =0
```

The contents of program counter before instruction, PC=address LAB_01

The contents of REG register after executing an instruction $REG = REG + 1$, if $REG = 0$, program counter points to label address LAB_03. Otherwise, program counter points to address of the next instruction or to LAB_02.

A.27 DECFSZ Decrement f, skip if = 0

Syntax: `[label] DECFSZ f, d`
Description: Contents of **f** register is decremented by one.
 If **d=0**, result is stored in **W** register.
 If **d=1**, result is stored in **f** register.
 If result = 0, next instruction is executed as NOP, thus making the current one, a two-cycle instruction.
Operation: $(f) - 1 \Rightarrow d$
Operand: $0 \leq f \leq 127$
 $d \in [0,1]$
Flag: -
Number of words: 1
Number of cycles: 1 or 2 depending on a result

Example

```
LAB_01      DECFSZ  CNT, 1           ; Decrement the contents REG by one.
LAB_02      . . . . .               ; Skip this line if = 0
LAB_03      . . . . .               ; Skip here if = 1
```

The contents of program counter before instruction, PC=address LAB_01

The contents of CNT register after executing an instruction CNT=CNT-1, if CNT=0, program counter points to address of label LAB_03. Otherwise, program counter points to address of the following instruction, or to LAB_02.

A.28 GOTO Jump to address

Syntax: `[label] GOTO k`
Description: Unconditional jump to address **k**.
Operation: $k \Rightarrow PC<10:0>, (PCLATH<4:3>) \Rightarrow PC<12:11>$
Operand: $0 \leq k \leq 2048$
Flag: -
Number of words: 1
Number of cycles: 2

Example

```
LAB_00      GOTO LAB_01           ; Jump to LAB_01
            :
            :
LAB_01      . . . . .
```

Before instruction: PC=address LAB_00
 After instruction: PC=address LAB_01

A.29 CALL Call a program

Syntax:	<i>[label]</i> CALL k
Description:	Instruction calls a subprogram. First, return address (PC+1) is stored on stack, then 11-bit direct operand k , which contains the subprogram address, is stored in program counter.
Operation:	(PC) + 1 \Rightarrow Top Of Stack (TOS) k \Rightarrow PC<10:0>, (PCLATH<4:3>) \Rightarrow PC<12:11>
Operand:	0 \leq k \leq 2048
Flag:	-
Number of words:	1
Number of cycles:	2

Example

```

LAB_01      CALL LAB_02                ; Call subroutine LAB_02
            :
            :
LAB_02      . . . . .

```

Before instruction: PC=address LAB_01
TOS=x

After instruction: PC=address LAB_02
TOS=LAB_01

A.30 RETURN Return from a subprogram

Syntax:	<i>[label]</i> RETURN
Description:	Contents from the top of a stack is stored in program counter.
Operation:	TOS \Rightarrow program counter PC
Operand:	-
Flag:	-
Number of words:	1
Number of cycles:	2

Example RETURN

Before instruction: PC=x
TOS=x

After instruction: PC=TOS
TOS=TOS-1

A.31 RETLW Return from a subprogram with constant in W

Syntax: `[label] RETLW k`
Description: 8-bit constant **k** is stored in **W** register. Value off the top of a stack is stored in program counter.
Operation: $(k) \Rightarrow W$; $TOS \Rightarrow PC$
Operand: $0 \leq k \leq 255$
Flag: -
Number of words: 1
Number of cycles: 2

Example `RETLW 0x43`

Before instruction: $W = X$
 $PC = X$
 $TOS = X$
 After instruction: $W = 0x43$
 $PC = TOS$
 $TOS = TOS - 1$

A.32 RETFIE Return from interrupt routine

Syntax: `[label] RETFIE`
Description: Return from a subprogram. Value from TOS is stored in program counter PC. Interrupts are enabled by setting a GIE (Global interrupt Enable) bit.
Operation: $TOS \Rightarrow PC$; $1 \Rightarrow GIE$
Operand: -
Flag: -
Number of words: 1
Number of cycles: 2

Example `RETFIE`

Before instruction: $PC = X$
 $GIE = 0$
 After instruction: $PC = TOS$

A.33 NOP No operation

Syntax: `[label] NOP`
Description: Does not execute any operation or affect any flag.
Operation: -
Operand: -
Flag: -
Number of words: 1
Number of cycles: 1

Example `NOP`

Before instruction: $PC = X$
 After instruction: $PC = X + 1$

A.34 CLRWD Initialize watchdog timer

Syntax:	[<i>label</i>] CLRWDT
Description:	Watchdog timer is reset. Prescaler of the Watchdog timer is also reset, and status bits \overline{TO} and \overline{PD} are set also.
Operation:	0 \Rightarrow WDT 0 \Rightarrow WDT prescaler 1 \Rightarrow \overline{TO} 1 \Rightarrow \overline{PD}
Operand:	-
Flag:	\overline{TO} , \overline{PD}
Number of words:	1
Number of cycles:	1

Example CLRWDT

Before instruction: WDT counter=x
WDT prescaler=1: 128

After instruction: WDT counter=0x00
WDT prescaler counter=0
 $\overline{TO}=1$
 $\overline{PD}=1$
WDT prescaler=1: 128

A.35 SLEEP Stand by mode

Syntax:	[<i>label</i>] SLEEP
Description:	Processor goes into low consumption mode. Oscillator is stopped. \overline{PD} (Power Down) status bit is reset. \overline{TO} (Timer Out) bit is set. WDT (Watchdog) timer and its prescaler are reset.
Operation:	0 \Rightarrow WDT 0 \Rightarrow WDT prescaler 1 \Rightarrow \overline{TO} 0 \Rightarrow \overline{PD}
Operand:	-
Flag:	\overline{TO} , \overline{PD}
Number of words:	1
Number of cycles:	1

Example SLEEP

Before instruction: WDT counter=x
WDT prescaler=x

After instruction: WDT counter=0x00
WDT prescaler=0
 $\overline{TO}=1$
 $\overline{PD}=0$

[Previous page](#)
[Table of contents](#)
[Chapter overview](#)
[Next page](#)

